A logarithmic amplifier SPICE macromodel improved with DC offset voltages and input bias currents

Ivailo M. Pandiev[1]

Abstract – **An improved SPICE macromodel of a logarithmic amplifier is presented here, in which the offset voltage and input bias current are modelled. The simulation model is developed through modifying the existing macromodels employing the mechanism of controlled sources and sub-circuits. The proposed macromodel allows simulating analogue circuits with respect to the behaviour in both the time and frequency domains, including error parameters and temperature dependence of the input (output) offset voltage and the input bias current. As an example, model parameters for the logarithmic amplifier LOG101 (Texas Instruments) are extracted. Simulated results and selected diagrams are compared with the manufacturer's data.**

Keywords **– Analogue circuits, Offset voltage and current, Input bias current, Logarithmic amplifier, SPICE, Modelling.**

I. INTRODUCTION

Logarithmic amplifiers (log amps) are widely used for compressing signals and computation. Although digital ICs have mostly replaced the log amps in applications that require computing, engineers continue to use log amps to compress signals. Therefore, the log amp remains an essential component in many video, medical, test and measurement systems. [1-3].

Accuracy considerations for log amps are somewhat more complicated than the ones for the other operational amplifiers. This is because the transfer function is nonlinear and log amps have two inputs, each varying in a wide dynamic range. The accuracy for any combination of the input voltages is determined from the total error. This is the deviation of the actual output voltage $V_{out(actual)}$ from the ideal output voltage $V_{out(ideal)}$. Thus,

$$V_{out(actual)} = V_{out(ideal)} \pm V_{error} \quad (1)$$

The $\pm V_{error}$ represents the sum of all the individual components of error normally associated with the logarithmic amplifier when operating in current input mode. As with any transfer function, the error generated by the function itself may be referred to the output or to the inputs. The individual error components of log amps are: input/output offset voltage, input bias and offset currents, as well as their temperature drifts. For most of the integrated log amps in voltage input mode of operation, the $V_{out(actual)}$ as a function of the major components of error can be found by

$$V_{out(actual)} = (1V)(1 \pm \Delta K)\,log\frac{\dfrac{V_1}{R_1} - I_{B1} \pm \dfrac{V_{OS1}}{R_1}}{\dfrac{V_2}{R_2} - I_{B2} \pm \dfrac{V_{OS2}}{R_2}} \pm 2Nn \pm V_{OSO} \quad (2)$$

where ΔK is gain error of the output voltage, I_{B1} and I_{B2} are the bias currents of input 1 and 2, V_{OS1} and V_{OS2} – input offset voltages, N – log conformity error (usually from 0,01% to 0,06%), n – number of decades and V_{OSO} – output offset voltage.

The offset voltage and bias currents of the log amps are important parameters, taken into consideration during the design of photodiode signal compression amplifiers, analogue signal compression before ADCs, etc. An accurate model of the log amp input bias current and input/output offset voltages must therefore take into account not only the voltage and current for *T*=25°C, normally quoted on data sheet, but also the temperature drifts in the operating range.

The majority of published SPICE log amp macromodels only attempt to model input bias current and offset voltage at *T*=25°C, usually by ideal DC sources, connected to the input terminals [4, 5]. They therefore cannot be used to simulate dynamic variations in input bias current and offset voltage in function with the change of temperature. In response to this problem, here is presented a log amp SPICE macromodel, with input/output offset voltages and input bias currents improved versus temperature. The proposed simulation macromodel is developed following the design procedure given in [7].

II. BUILDING THE MACROMODELS SENSING INPUT AND OUTPUT OFFSETS

The schematic diagram of the proposed macromodel is shown in Fig. 1. It is based on a previous SPICE macromodel [4], with the additional capability to simulate the input/output offset voltage and the input bias/offset current versus temperature. In Fig. 1, there are two operational amplifier models represented as black-boxes, described in full detail in [6].

Various parameters have been modelled as follows.

A. Modelling the input and output offset voltages

These parameters are modelled by adding new elements and stages to the equivalent circuit of the existing macromodel of the log amp. As it is shown in Fig. 1, two linear voltage-controlled voltages sources (VCVSs) E_{OS1} and E_{OS2} are

[1]Ivailo M. Pandiev is with the Faculty of Electronics from Technical University of Sofia, Kliment Ohridski 8, 1000 Sofia, Bulgaria. E-mail: ipandiev@tu-sofia.bg

connected to the non-inverting inputs of the amplifiers, creating the new nodes 161 and 162. The state of those nodes will have to follow the change in the input offset voltage. Node 16 through $R_{PI} = 10\Omega$ is connected to node 6, which is the reference node of the macromodel.

The current I_{TVOS} (and I_{TVOSO}) will flow through the resistor R_{TVOS} (and R_{TVOSO}), towards the internal ground. In such a way the voltage V_{100} (and V_{101}) will depend upon the temperature.



Fig. 1. Conceptual circuit of the proposed log amp SPICE macromodel.

The output offset voltage is modelled with one VCVS E_{OSO}, connected between nodes 3 (output terminal) and 331. The temperature-dependent controlling voltages of the E_{OS1}, E_{OS2} and E_{OSO} come from separate temperature stages in Fig. 1. These additionally defined stages consist of ideal current sources (I_{TVOS} and I_{TVOSO}) and SPICE temperature-dependent resistors (R_{TVOS} and R_{TVOSO}), controlled with the equation [8]:

$$R\ (T) = R(T_0)\left[1 + TC1(T - T_0) + TC2(T - T_0)^2\right] \quad (3)$$

where $R(T_0)$ is the value of the resistor at $T_0 = 27°C$ (SPICE-Option TNOM), T is the temperature in $°C$, $TC1$ is the linear temperature coefficient and $TC2$ is the quadratic temperature coefficient. A sufficient degree of accuracy for the purpose of modelling offset voltage temperature drift has been provided by choosing $R(T)$ as a linear temperature-dependent resistor ($TC2 = 0$), having the following characteristic equation:

$$R(T) = R(T_0)[1 + TC1(T - T_0)] \quad (4)$$

$$V_{100,0} = I_{TVOS} R_{TVOS}(T_0)[1 + TC1(T - T_0)] \quad (5a)$$

$$V_{101,0} = I_{TVOSO} R_{TVOSO}(T_0)[1 + TC1(T - T_0)] \quad (5b)$$

The parameters of the current I_{TVOS} (and I_{TVOSO}) and the resistor R_{TVOS} (and R_{TVOSO}) are calculated so that the temperature stages provide a voltage $V_{100} = V_{101} = 1mV$ at the temperature $T_0 = 27°C$. Signals, generated at nodes 100 and 101, are used for forming the characteristic equations of E_{OS1}, E_{OS2} and E_{OSO} as follows:

$$V(E_{OS1}) = k_{o,EOS1} + k_{1,EOS1} V_{100}, \quad (6a)$$

$$V(E_{OS2}) = k_{o,EOS2} + k_{1,EOS2} V_{100}, \quad (6b)$$

$$V(E_{OSO}) = k_{o,EOSO} + k_{1,EOSO} V_{102}. \quad (6c)$$

The linear temperature coefficient $TC1$ from Eq. (4), and characteristic coefficients from Eq. (6a), (6b) and (6c) are calculated by employing the least squares sense technique.

B. Modelling the input bias current and the input offset current

The analysis of data sheets for the log amps has shown that in most cases the input characteristics could be approximated by nonlinear functions, in which input bias currents double with every 10°C temperature rise. Normally, semiconductor data books, provided by IC-manufacturers, contain information regarding the temperature dependence of the input currents that could be used in modelling the real log amps.

The input bias/offset currents are implemented by non-linear voltage-controlled current sources (VCCSs) G_{IBn}, G_{IBp} and G_{IO}, connected to the inverting and non-inverting inputs of the amplifiers and an additionally defined temperature stage consisting of an ideal current source I_{TB} and a temperature-dependent resistor R_{TB}, as shown in Fig. 1.

The temperature stage provides a voltage V_{102} for modelling linear temperature dependences using the SPICE model (4) of the linear temperature-dependent resistor ($TC2 = 0$). For convenience, the parameters of the current I_{TB} and the resistor R_{TB} are extracted so that the temperature stage provides a voltage $V_{102} = 1V$ at the temperature $T_0 = 27°C$. Since SPICE-simulators will give an error message if the resistor R_{TB} goes negative at any temperature, the value of the linear coefficient $TC1$ can be obtained by

$$TC1 \le 10^{-1} / (T_0 - T_{\min}) \tag{7}$$

where T_{min} is the minimum temperature in the operating range of the real logarithmic amplifier.

The controlling voltage V_{102} is a linear temperature-controlled input value for the VCCSs G_{IBn}, G_{IBp} and G_{IO} in the model. The approximation of the non-linear behaviour is realized using appropriate polynomial coefficients of the controlled sources. A sufficient degree of accuracy for the purpose of modelling temperature dependence has been provided by choosing the polynomial sources realized by sixth-order functions. The VCCS functions are given by

$$I(G_{IBn}) = \sum_{i=0}^{n=6} k_{i,IBn} V_{IBn}^{i} , \tag{8a}$$

$$I(G_{IBp}) = \sum_{i=0}^{n=6} k_{i,IBp} V_{IBp}^{i} , \tag{8b}$$

$$I(G_{IO}) = \sum_{i=0}^{n=6} k_{i,IO} V_{IO}^{i} . \tag{8c}$$

The polynomial coefficients ($k_{i,IBn}, k_{i,IBp}$ and $k_{i,IO}$ $i = \overline{0,n}$) are calculated using MATLAB. In particular, the function *polyfit.m* is implemented to define the unknown coefficients of the controlled sources. In general, the function *polyfit* (*x,y,n*) for a given data in a vector $x\,(T,°C)$ finds a *n*-th order polynomial function *p* such that $p(x)$ fits the data in a vector *y* (positive and negative input bias current) by using the least squares sense. The main advantage of this approach is that parameter extraction can be performed only from data sheet parameters, even for circuits whose internal structure is unknown.

III. MACROMODEL PERFORMANCE

The verification of the SPICE macromodel (Fig. 1) with temperature-dependent input and output offset voltages and currents is carried out by comparing simulation results with data sheet parameters of integrated log amp LOG101 [9]. The sub-circuit listing, given in Table 1, contains elements (marked in bold style) for simulating the DC voltage and current error of the LOG101. The test circuit for simulation is created following the test conditions given in the semiconductor data book of the corresponding IC. The power supply voltages of the circuit are chosen to be ±15V. The simulation modelling is implemented within EDA OrCAD. During the process of simulation a DC sweep analysis is specified and performed with the temperature ranging from -5°C to 75°C. In the Fig. 2, 3, 4 and 5 are presented the simulation output ($x_{,M}$), the data sheet parameters (x_{DS}) for the input/output offset voltages, input bias/offset currents and error ($\delta = [(x_M - x_{DS}) / x_M].100\%$) as a function of the temperature. It can be observed, that the accuracy of the macromodel is quite good ($\delta < 5\%$).

Table 1. SPICE netlist for LOG101 macromodel

```
* PINOUT ORDER I1 VOUT V+ V- GND I2
* PINOUT ORDER  1   3   4  5  6   8
.SUBCKT LOG101 1 3 4 5 6 8
R1 9 331 3340
R2 6 9 212 TC1=3.47M
Q1 10 9 11 QL101
Q2 12 6 11 QL101
X1 162 8 4 5 3 LA101
X2 161 1 4 5 13 LA101
* Input and output offset voltages
EOSO 3 331 101 0 3.00386666666667
EOS1 162 16 100 16 0.2663
EOS2 161 16 100 16 0.2663
* Input bias and offset currents
GIBn 1 0 POLY(1) (102 0) 0.007518015195925
+-0.04450102859145  0.109768095815289  -0.144422090643820  +0.10689777201416
-0.042204603929841 0.006943840145432
GIBp 161 0 POLY(1)   (102 0) 0.007518015195925
+-0.04450102859145 0.109768095815289 -0.144422090643820 +0.106897772014167
-0.042204603929841 0.006943840145432
GIO 1 161 POLY(1) (102 0) 0.0007518015195925
+-0.0044501028591453 0.0109768095815287
+-0.0144422090643818 0.0106897772014165
+ -0.0042204603929840 0.0006943840145432
* Temperature stage for input offset voltage
ITVOS 0 100 1m
RTVOS 100 0 1 TC1=0.00813516
* Temperature stage for output offset voltage
ITVOSO 0 101 1m
RTVOSO 101 0 1 TC1=6.436148963546978e-004
* Temperature stage for input bias and offset current
ITB 0 102 1
RTB 102 0 1 TC1=0.001
RPI 16 6 10
R5 0 8 1E12
R6 0 1 1E12
R7 14 11 670
C1 1 13 154E-12
D3 14 13 DS
R8 14 13 1E8
R9 11 1 1E12
R10 8 11 1E12
R11 8 12 82
R12 10 1 82
.ENDS
.SUBCKT LA101 3 2 7 4 6
* SPICE netlist for LA101 ***see Reference [6] ***
.ENDS
* END MODEL LOG101
```

For the macromodel, the simulation time is observed to be comparable to the existing models from the standard SPICE libraries. At present, the advantage of such *complex* macromodels seems to be the study of the effects of specific component parameters of the complete electronic sub-systems.



Fig. 2. Input offset voltage as a function of temperature for LOG101



Fig. 3. Output offset voltage as a function of temperature for LOG101



Fig. 4. Input bias current as a function of temperature for LOG101



Fig. 5. Input offset current as a function of temperature for LOG101

IV. Acknowledgment

The author wish to thank the reviewers who offered useful and constructive comments.

V. Conclusions

This paper presents the new developments of log amp simulation macromodel, aimed at improving the modelling of the offset voltage and input bias current versus temperature. The model is independent from actual technical realizations and is based on compromises regarding the exact circuit structure in the model. The efficiency of the model was proved by comparison of simulation results and data sheet parameters of the integrated logarithmic amplifier LOG101 from Texas Instruments.

References

[1] R. L. Geiger, P. E. Allen and N. R. Strader, *VLSI Design Techniques for Analog and Digital Circuits*, pp. 289-302, McGraw-Hill, 1996.

[2] Tietze, U., Ch. Schenk. *Halbleiter-Schaltungstechnik*. 11.1 Auflage. Springer-Verlag, pp. 785-800, 2001.

[3] *Integrated dc logarithmic amplifiers*, AN 3611, Maxim-Dallas Semiconductor, 2005, http://maxim-ic.com.

[4] *Simulation models – logarithmic amplifiers*, Texas Instruments, 2005, http://www.ti.com.

[5] S. R. Chowdhury, C. Pramanik, H. Saha, "ASIC Design of the Linearisation Circuit of a PTC Thermistor", *18th Int. Conf. on VLSI Design'05*, pp. 866-869, 2005.

[6] Boyle G., B. Cohn. D. Pederson, J. Solomon. "Macromodeling of Integrated Circuit Operational Amplifiers". *IEEE Journal of Solid State Circuits*, Vol. SC-9, No.6, 1974, pp. 353- 363.

[7] I.M. Pandiev, "Model building and testing procedure - An analogue multiplexer SPICE macromodel improved with temperature effects," ICEST 2006 (submitted).

[8] *MicroSim PSpice A/D Reference Manual*, MicroSim, Irvine, California USA, Chapter 2, 1996.

[9] *Precision logarithmic amplifier LOG101* - data sheet, Texas Instruments, USA, 2002.